// SPDX-License-Identifier: MIT
/*
(\_/)
( •_•)
/ > Basic Buns

by Christian Montoya
*/
pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}


/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}



/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}




/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}




/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}





/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) private pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}



/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}




/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        return string(abi.encodePacked("Bunny Hold Creature"));
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        require(operator != _msgSender(), "ERC721: approve to caller");

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer of token that is not owned");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver(to).onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}




interface EmojiGardenInterface {
    function ownerOf(uint256 tokenId) external view returns (address owner);
}


contract BasicBuns is ERC721, Ownable {

    uint256 private _maxSupply = 1000; // most we will ever have - this number can be lowered but not raised
    uint256 private _basePrice = 25000000; // 0.025 ether - this number can be changed 
    uint256 private payableId = 416; // payableId + 1 = id of next token you can buy 
    uint256 private claimedAmount = 0; // increment whenever a token is claimed with a garden 
    
    // EmojiGardenInterface public emojiGardenContract = EmojiGardenInterface(0x4a15C2d191D70722552d8F67B61A36784E9d475D); // testnet 
    EmojiGardenInterface public emojiGardenContract = EmojiGardenInterface(0x690dBdf6bB1712f01b34e80c25553DB869dF8Bf9); // mainnet 

    struct bunnyData {
        string colorSwatchId;
        string colorSwatchOne;
        string colorSwatchTwo;
        string backgroundId;
        string background;
        string earId;
        string ears;
        string faceId;
        string face;
        string bodyId;
        string body;
        string animal;
        string emoji;
    }

    string[9] private _colorSwatchIds = [
        "Dusk",
        "Midnight",
        "Stealth",
        "Fire",
        "Blood",
        "Beauty",
        "Deep",
        "Slate",
        "Neon"
    ];

    string[9] private _colorSwatchOnes = [
        "#80F",
        "#306",
        "#026",
        "#e83",
        "#a00",
        "#d18",
        "#1bb",
        "#456",
        "#0cc"
    ];

    string[9] private _colorSwatchTwos = [
        "#f40",
        "#a4c",
        "#002",
        "#e52",
        "#300",
        "#300",
        "#303",
        "#123",
        "#f1f"
    ];

    string[7] private _backgroundIds = [
        "Blank",
        "Dots",
        "Chevrons",
        "Triangles",
        "Squares",
        "Diamonds",
        "Bricks"
    ];

    string[7] private _backgrounds = [
        '<pattern id="b" x="0" y="0" width="100" height="100"><rect fill="#fff" fill-opacity="0" width="100" height="100"></rect></pattern>',
        '<pattern id="b" x="0" y="0" width="30" height="30" patternUnits="userSpaceOnUse"><g fill="#fff" opacity=".15"><circle cx="15" cy="15" r="3"/><circle cy="30" r="3"/><circle cx="30" cy="30" r="3"/><circle cx="30" r="3"/><circle r="3"/></g></pattern>',
        '<pattern id="b" x="0" y="0" width="24" height="24" patternUnits="userSpaceOnUse"><polygon fill="#fff" opacity="0.15" points="24 24 12 24 18 18 24 12 24 0 24 0 12 12 0 0 0 12 6 18 12 24 24 24 "/></pattern>',
        '<pattern id="b" x="0" y="0" width="12" height="12" patternUnits="userSpaceOnUse"><polygon fill="#fff" opacity="0.15" points="0 0 12 0 0 12"/></pattern>',
        '<pattern id="b" x="0" y="0" width="40" height="40" patternUnits="userSpaceOnUse"><rect fill="#fff" opacity="0.15" x="0" y="0" width="20" height="20"/><rect fill="#fff" opacity="0.15" x="20" y="20" width="20" height="20"/></pattern>',
        '<pattern id="b" x="10" y="10" width="20" height="20" patternUnits="userSpaceOnUse"><circle fill="#fff" cx="10" cy="10" r="10" opacity="0.15"/></pattern>',
        '<pattern id="b" x="0" y="0" width="42" height="42" patternUnits="userSpaceOnUse"><g fill="none" fill-rule="evenodd"><g fill="#fff" opacity="0.15"><path d="M0 0h42v44H0V0zm1 1h40v20H1V1zM0 23h20v20H0V23zm22 0h20v20H22V23z"/></g></g></pattern>'
    ];

    string[3] private _earIds = [
        'Pointy',
        'Furry',
        'Kitty'
    ];

    string[3] private _ears = [
        '(\\__/)',
        '{\\__/}',
        '/\\_/\\'
    ];

    string[18] private _faceIds = [
        'Calm',
        'Puzzled',
        'Kitty',
        'Mischievous',
        'Happy',
        'Concerned',
        'Timid',
        'Confident',
        'Derpy',
        'Loud',
        'Disappointed',
        'Smug',
        'Skeptical',
        'Mysterious',
        'Rabid',
        'Victorious',
        'Desperate',
        'Human'
    ];

    string[18] private _faces = [
        unicode"(  •_•)",
        unicode"( • . •)",
        unicode"(=•ㅅ•)",
        unicode"( • - •)",
        unicode"( ^_^)",
        unicode"(⚆_⚆)",
        unicode"(  ' . ')",
        unicode"(˵ 'ω')",
        unicode"( • ∇ •)",
        unicode"( ° □ °)",
        unicode"( T_T)",
        unicode"(  ᵔᴥᵔ)",
        unicode"(  •̃_•̃ )",
        unicode"( ◕｡◕)",
        unicode"ミ●˾☉",
        unicode"(  •̀ᴗ•́ )",
        unicode"(｡•́‿•̀)",
        unicode"( ಠʖ̯ಠ)"
    ];

    string[3] private _bodyIds = [
        'Forward',
        'Paw',
        'Back'
    ];

    string[3] private _bodies = [
        unicode"/ &gt;",
        unicode"/づ",
        unicode"&lt; \\"
    ];

    string[3] private _bodiesRaw = [
        unicode"/ >",
        unicode"/づ",
        unicode"< \\"
    ];

    string[111] private _emojis = [
        unicode"💀",
        unicode"💩",
        unicode"🧳",
        unicode"🎃",
        unicode"🧵",
        unicode"🧶",
        unicode"🧦",
        unicode"🩲",
        unicode"🎩",
        unicode"💍",
        unicode"🐣",
        unicode"🐌",
        unicode"💐",
        unicode"🌸",
        unicode"🌹",
        unicode"🍀",
        unicode"🍄",
        unicode"🦀",
        unicode"🌎",
        unicode"🌞",
        unicode"🌟",
        unicode"🌪️",
        unicode"🌈",
        unicode"⛄",
        unicode"🔥",
        unicode"🎄",
        unicode"🍉",
        unicode"🍍",
        unicode"🍎",
        unicode"🍑",
        unicode"🍒",
        unicode"🍆",
        unicode"🍞",
        unicode"🧀",
        unicode"🍕",
        unicode"🌮",
        unicode"🍿",
        unicode"🍙",
        unicode"🍣",
        unicode"🥡",
        unicode"🍦",
        unicode"🍪",
        unicode"🎂",
        unicode"🧁",
        unicode"🍼",
        unicode"🍵",
        unicode"🍹",
        unicode"🍺",
        unicode"🥤",
        unicode"🧊",
        unicode"🤸",
        unicode"🏆",
        unicode"⚽",
        unicode"⚾",
        unicode"🏀",
        unicode"🏓",
        unicode"🎣",
        unicode"🎱",
        unicode"🎮",
        unicode"🎲",
        unicode"🧩",
        unicode"🎨",
        unicode"🎤",
        unicode"🎸",
        unicode"🥁",
        unicode"🎬",
        unicode"🌋",
        unicode"🚘",
        unicode"🚨",
        unicode"🪐",
        unicode"🚀",
        unicode"💌",
        unicode"💣",
        unicode"🔪",
        unicode"🧭",
        unicode"🛎️",
        unicode"⏱️",
        unicode"🧨",
        unicode"🎉",
        unicode"🧧",
        unicode"🎁",
        unicode"🕹️",
        unicode"🧸",
        unicode"💎",
        unicode"📻",
        unicode"📱",
        unicode"📟",
        unicode"💻",
        unicode"📸",
        unicode"📹",
        unicode"🔎",
        unicode"💡",
        unicode"🏮",
        unicode"📓",
        unicode"📖",
        unicode"🗞️",
        unicode"💰",
        unicode"💳",
        unicode"📦",
        unicode"🗑️",
        unicode"🔑",
        unicode"🛡️",
        unicode"🧲",
        unicode"🧪",
        unicode"💉",
        unicode"💊",
        unicode"🧹",
        unicode"🧻",
        unicode"🧯",
        unicode"💝",
        unicode"🔔"
    ];

    function memcmp(bytes memory a, bytes memory b) internal pure returns(bool){
        return (a.length == b.length) && (keccak256(a) == keccak256(b));
    }
    function strcmp(string memory a, string memory b) internal pure returns(bool){
        return memcmp(bytes(a), bytes(b));
    }
    
    function totalSupply() public view virtual returns (uint256) { 
        uint256 total = claimedAmount + (payableId-416); 
        return total; 
    }
    
    function makeBunny(uint256 tokenId) internal view returns (bunnyData memory) { 
        bunnyData memory bunny;

        // random background gradient
        uint256 rand = uint256(keccak256(abi.encodePacked(tokenId, address(this), "4"))) % _colorSwatchIds.length;
        bunny.colorSwatchId = _colorSwatchIds[rand];
        bunny.colorSwatchOne = _colorSwatchOnes[rand];
        bunny.colorSwatchTwo = _colorSwatchTwos[rand];

        // random background pattern
        rand = uint256(keccak256(abi.encodePacked(tokenId, address(this), "5"))) % _backgroundIds.length;
        bunny.backgroundId = _backgroundIds[rand];
        bunny.background = _backgrounds[rand];

        // 3 types of ears, differing rarity
        rand = uint256(keccak256(abi.encodePacked(tokenId, address(this), "1"))) % 7;
        bunny.earId = _earIds[0];
        bunny.ears = _ears[0];
        if(rand > 5) {
            bunny.earId = _earIds[2];
            bunny.ears = _ears[2];
        }
        else if(rand > 3) {
            bunny.earId = _earIds[1];
            bunny.ears = _ears[1];
        }

        // 17 types of faces, also differing rarity
        uint256 faceRange = 8;
        rand = uint256(keccak256(abi.encodePacked(tokenId, address(this)))) % 7;
        if(rand >= 5) {
            faceRange = 18;
        }
        else if(rand >= 2) {
            faceRange = 13;
        }
        rand = uint256(keccak256(abi.encodePacked(tokenId, address(this), "2"))) % faceRange;
        bunny.faceId = _faceIds[rand];
        bunny.face = _faces[rand];
        delete faceRange;

        // use face to determine if kitty or human
        bunny.animal = "Bunny";
        if(strcmp(bunny.faceId, "Human")) {
            bunny.animal = "Human";
        }
        else if(strcmp(bunny.earId,"Kitty")) {
            bunny.animal = "Kitty";
        }

        // 3 types of bodies, also differing rarity
        rand = uint256(keccak256(abi.encodePacked(tokenId, address(this), "3"))) % 9;
        bunny.bodyId = _bodyIds[0];
        bunny.body = _bodies[0];
        if(rand >= 8) {
            bunny.bodyId = _bodyIds[2];
            bunny.body = _bodies[2];
        }
        else if(rand >= 6) {
            bunny.bodyId = _bodyIds[1];
            bunny.body = _bodies[1];
        }

        // lots of possible emoji
        rand = uint256(keccak256(abi.encodePacked(address(this), tokenId))) % _emojis.length;
        bunny.emoji = _emojis[rand];
        
        return bunny; 
    }
    
    function makeSVG(bunnyData memory bunny) internal pure returns (string memory) { 
        string[14] memory parts;

        parts[0] = '<svg xmlns="http://www.w3.org/2000/svg" preserveAspectRatio="xMinYMin meet" viewBox="0 0 200 200" width="100%" height="100%">';
        parts[1] = "<style>text{font-size:32pt;font-family:'Helvetica Neue',Arial,sans-serif;font-weight:bold;text-anchor:middle}.tt{fill:#fff;white-space:pre}#stop1{stop-color:";
        parts[2] = bunny.colorSwatchOne;
        parts[3] = '}#stop2{stop-color:';
        parts[4] = bunny.colorSwatchTwo;
        parts[5] = "}</style><defs><linearGradient id='a' x1='0' x2='0' y1='0' y2='1'><stop offset='0' id='stop1' /><stop offset='1' id='stop2' /></linearGradient></defs><rect width='100%' height='100%' fill='url(#a)'/>";
        parts[6] = bunny.background;
        parts[7] = "<rect width='100%' height='100%' fill='url(#b)'/><text class='tt' x='50%' y='28%'>";
        parts[8] = bunny.ears;
        parts[9] = "</text><text class='tt' x='50%' y='56%'>";
        parts[10] = bunny.face;
        parts[11] = "</text><text x='50%' y='87%'>";
        parts[12] = bunny.body;
        parts[13] = "</text></svg>";

        string memory output = string(abi.encodePacked(parts[0], parts[1], parts[2], parts[3], parts[4], parts[5], parts[6]));
        output = string(abi.encodePacked(output, parts[7], parts[8], parts[9], parts[10], parts[11], parts[12], parts[13]));

        return output;   
    }
    
    function bunnyString(uint256 tokenId) public view returns (string memory) {
        require(ownerOf(tokenId) != address(0), "ERC721: bunnyString query for nonexistent token");

        bunnyData memory bunny = makeBunny(tokenId);
        
        if(strcmp(bunny.bodyId,"Back")) {
            bunny.body = unicode"< \\"; 
            bunny.body = string(abi.encodePacked(bunny.emoji, bunny.body));
        }
        else if(strcmp(bunny.bodyId,"Forward")) { 
            bunny.body = unicode"/ >";
            bunny.body = string(abi.encodePacked(bunny.body, bunny.emoji));
        }
        else {
            bunny.body = string(abi.encodePacked(bunny.body, bunny.emoji));
        }

        return string(abi.encodePacked(bunny.ears, "\r\n", bunny.face, "\r\n", bunny.body));
    }
    
    function bunnySVG(uint256 tokenId) public view returns (string memory) { 
        require(ownerOf(tokenId) != address(0), "ERC721: bunnySVG query for nonexistent token");

        bunnyData memory bunny = makeBunny(tokenId);
        
        if(strcmp(bunny.bodyId,"Back")) {
            bunny.body = string(abi.encodePacked(bunny.emoji, "<tspan class='tt'>", bunny.body, "</tspan>"));
        }
        else {
            bunny.body = string(abi.encodePacked("<tspan class='tt'>", bunny.body, "</tspan>", bunny.emoji));
        }

        string memory output = makeSVG(bunny); 

        return output;   
    }

    function tokenURI(uint256 tokenId) override public view returns (string memory) {
        require(ownerOf(tokenId) != address(0), "ERC721: tokenURI query for nonexistent token");

        bunnyData memory bunny = makeBunny(tokenId);  
        
        if(strcmp(bunny.bodyId,"Back")) {
            bunny.body = string(abi.encodePacked(bunny.emoji, "<tspan class='tt'>", bunny.body, "</tspan>"));
        }
        else {
            bunny.body = string(abi.encodePacked("<tspan class='tt'>", bunny.body, "</tspan>", bunny.emoji));
        }

        string memory output = makeSVG(bunny); 
        
        string[15] memory parts;

        parts[0] = '[{"trait_type":"Pattern","value":"';
        parts[1] = bunny.backgroundId;
        parts[2] = '"},{"trait_type":"Backdrop","value":"';
        parts[3] = bunny.colorSwatchId;
        parts[4] = '"},{"trait_type":"Creature","value":"';
        parts[5] = bunny.animal;
        parts[6] = '"},{"trait_type":"Ears","value":"';
        parts[7] = bunny.earId;
        parts[8] = '"},{"trait_type":"Face","value":"';
        parts[9] = bunny.faceId;
        parts[10] = '"},{"trait_type":"Body","value":"';
        parts[11] = bunny.bodyId;
        parts[12] = '"},{"trait_type":"Holding","value":"';
        parts[13] = bunny.emoji;
        parts[14] = '"}]';

        string memory json = string(abi.encodePacked(parts[0], parts[1], parts[2], parts[3], parts[4], parts[5], parts[6], parts[7]));

        json = string(abi.encodePacked(json, parts[8], parts[9], parts[10], parts[11], parts[12], parts[13], parts[14]));
        
        if(strcmp(bunny.animal, "Human")) {
        }
        else if(strcmp(bunny.earId,"Kitty") && strcmp(bunny.faceId,"Kitty")) { 
            bunny.animal = "Kitty Cat";    
        }
        else { 
            bunny.animal = string(abi.encodePacked(bunny.faceId, " ", bunny.animal));    
        }

        bunny.animal = string(abi.encodePacked(bunny.animal, ' #', toString(tokenId)));

        json = string(abi.encodePacked('{"name":"',bunny.animal,'", "description":"A cute creature generated and stored entirely on-chain.","attributes":',json));
        json = Base64.encode(bytes(string(abi.encodePacked(json, ', "image": "data:image/svg+xml;base64,', Base64.encode(bytes(output)), '"}'))));
        output = string(abi.encodePacked('data:application/json;base64,', json));

        return output;
    }
    
    function ownerSetPrice(uint256 priceValue) public onlyOwner { 
        _basePrice = priceValue; 
    }
    
    function mint() public payable { 
        require(msg.value >= _basePrice, "Payable amount is incorrect, please check mint price"); 
        require(_maxSupply > payableId, "Public sale is over");
        payableId = payableId + 1; 
        _safeMint( msg.sender, payableId ); 
        (bool success,) = owner().call{value: msg.value}("");
        require(success);
    }
    
    function mintMulti(uint256 amount) public payable { 
        require(_maxSupply >= (payableId + amount), "Remaining supply too low to multi-mint");
        require(msg.value >= (_basePrice * amount), "Payable amount is incorrect, please check mint price"); 
        for(uint256 i = 0; i < amount; i++) { 
            payableId = payableId + 1; 
            _safeMint( msg.sender, payableId ); 
        }
    }
    
    function claimWithGarden(uint256 tokenId) public {
        require(tokenId > 0 && tokenId <= 416, "Token ID invalid");
        require(_maxSupply > totalSupply(), "Mint is over");
        require(emojiGardenContract.ownerOf(tokenId) == msg.sender, "You need to own the Emoji Garden with this ID to claim this Basic Bun");
        claimedAmount = claimedAmount + 1; 
        _safeMint(_msgSender(), tokenId);
    }
    
    function claimWithHonoraryGarden(uint256 tokenId) public { 
        require(tokenId > 4420 && tokenId <= 4444, "Token ID invalid"); 
        require(_maxSupply > totalSupply(), "Mint is over");
        require(emojiGardenContract.ownerOf(tokenId) == msg.sender, "You need to own the Emoji Garden with this ID to claim this Basic Bun");
        tokenId = tokenId - 4054; // 4421 maps to 367, 4444 maps to 390 
        claimedAmount = claimedAmount + 1;
        _safeMint(_msgSender(), tokenId);
    }
    
    function claimToWithGarden(address recipient, uint256 tokenId) public { 
        require(tokenId > 0 && tokenId <= 416, "Token ID invalid");
        require(_maxSupply > totalSupply(), "Mint is over");
        require(emojiGardenContract.ownerOf(tokenId) == msg.sender, "You need to own the Emoji Garden with this ID to gift this Basic Bun");
        claimedAmount = claimedAmount + 1; 
        _safeMint(address(recipient), tokenId);
    }

    function multiClaimWithGardens(uint256[] memory tokenIds) public {
        require(_maxSupply > totalSupply(), "Mint is over");
        for (uint256 i = 0; i < tokenIds.length; i++) {
            require(tokenIds[i] > 0 && tokenIds[i] <= 416, "Token ID invalid");
            require(emojiGardenContract.ownerOf(tokenIds[i]) == msg.sender, "You need to own the Emoji Garden with this ID to claim this Basic Bun");
            claimedAmount = claimedAmount + 1; 
            _safeMint(_msgSender(), tokenIds[i]);
        }
    }

    function withdraw() public onlyOwner(){
        payable(msg.sender).transfer(address(this).balance);
    }

    function maxSupply() public view returns (uint256) {
        return _maxSupply;
    }
    
    function basePrice() public view returns (uint256) {
        return _basePrice;
    }

    function reduceSupply(uint256 value) public onlyOwner(){
        require(value >= totalSupply() && value < _maxSupply, "Value not in required bounds");
        _maxSupply = value;
    }

    function endMint() public onlyOwner(){
        require(_maxSupply != totalSupply(), "Mint is already over");
        _maxSupply = totalSupply(); // this kills the mint
    }

    function toString(uint256 value) internal pure returns (string memory) {
    // Inspired by OraclizeAPI's implementation - MIT license
    // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    constructor() ERC721("Basic Buns", "BSCBN") Ownable() {}
}

/// [MIT License]
/// @title Base64
/// @notice Provides a function for encoding some bytes in base64
/// @author Brecht Devos <brecht@loopring.org>
library Base64 {
    bytes internal constant TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /// @notice Encodes some bytes to the base64 representation
    function encode(bytes memory data) internal pure returns (string memory) {
        uint256 len = data.length;
        if (len == 0) return "";

        // multiply by 4/3 rounded up
        uint256 encodedLen = 4 * ((len + 2) / 3);

        // Add some extra buffer at the end
        bytes memory result = new bytes(encodedLen + 32);

        bytes memory table = TABLE;

        assembly {
            let tablePtr := add(table, 1)
            let resultPtr := add(result, 32)

            for {
                let i := 0
            } lt(i, len) {

            } {
                i := add(i, 3)
                let input := and(mload(add(data, i)), 0xffffff)

                let out := mload(add(tablePtr, and(shr(18, input), 0x3F)))
                out := shl(8, out)
                out := add(out, and(mload(add(tablePtr, and(shr(12, input), 0x3F))), 0xFF))
                out := shl(8, out)
                out := add(out, and(mload(add(tablePtr, and(shr(6, input), 0x3F))), 0xFF))
                out := shl(8, out)
                out := add(out, and(mload(add(tablePtr, and(input, 0x3F))), 0xFF))
                out := shl(224, out)

                mstore(resultPtr, out)

                resultPtr := add(resultPtr, 4)
            }

            switch mod(len, 3)
            case 1 {
                mstore(sub(resultPtr, 2), shl(240, 0x3d3d))
            }
            case 2 {
                mstore(sub(resultPtr, 1), shl(248, 0x3d))
            }

            mstore(result, encodedLen)
        }

        return string(result);
    }
}